

## PERCOBAAN

### 3

#### JUDUL :


#### KONTROL DISPLAY 7-SEGMENT Menggunakan DT-51 MinSys

#### TUJUAN :


- Membuat proses kontrol display seven segment menggunakan SWITCH sebagai set (H) dan reset (L) program.
- Membuat program Counter baik Up Counter maupun Down Counter dengan media tampilan 7-Segment.

#### TEORI :

Sebagai media display/tampilan seven segment memiliki identifikasi segment seperti ditunjukkan pada gambar 3-1.a. Dengan identifikasi tersebut sangat membantu kita dalam membuat program untuk menampilkan nyala LED 7-Segment mulai dari hitungan decimal 0 sampai dengan decimal 15 seperti diilustrasikan pada gambar dibawah (gambar 3-1.b).




a) Identifikasi segment



b) Hasil tampilan pada seven segment

Gambar 3-1. Identifikasi 7-Segment

7-Segment merupakan kumpulan dari beberapa dioda, yaitu dioda dengan identifikasi a, b, c, d, e, f, g dan dp (Decimal Point). Pada gambar 3-2 ditunjukkan rangkaian dioda yang membentuk 7-segment.



Gambar 3-2. Rangkaian dioda pada 7-Segment

Dari gambar 3-2 dapat diketahui bahwa rangkaian dioda pembentuk 7-segment mempunyai tipe Common Anode, dimana anoda dari dioda tersebut secara bersama dihubungkan dengan catu daya (positip tegangan supply). Sedangkan untuk menyalakan led/dioda tipe ini katoda-nya biasanya dihubungkan dengan ground(GND).

Untuk mendapatkan hasil tampilan angka decimal seperti pada gambar diatas (3-1.b) diperlukan sebuah driver 7-Segment yang berfungsi untuk mengkonversi dari BCD ke seven segment menggunakan sebuah IC 74LS47. Tabel dari driver BCD ke 7-Segment adalah sebagai berikut :

Tabel 3-1. Driver BCD ke 7-Segment (IC 74LS47)

Decimal	Input					BI/ RBO	Output							
	LT	RBI	D	C	B		A	a	b	c	d	e	f	g
0	H	H	L	L	L	L	H	1	1	1	1	1	1	0
1	H	X	L	L	L	H	H	0	1	1	0	0	0	0
2	H	X	L	L	H	L	H	1	1	0	1	1	0	1
3	H	X	L	L	H	H	H	1	1	1	1	0	0	1
4	H	X	L	H	L	L	H	0	1	1	0	0	1	1
5	H	X	L	H	L	H	H	1	0	1	1	0	1	1
6	H	X	L	H	H	L	H	0	0	1	1	1	1	1

7	H	X	L H H H	H	1	1	1	0	0	0	0
8	H	X	H L L L	H	1	1	1	1	1	1	1
9	H	X	H L L H	H	1	1	1	0	0	1	1
10	H	X	H L H L	H	0	0	0	1	1	0	1
11	H	X	H L H H	H	0	0	1	1	0	0	1
12	H	X	H H L L	H	0	1	0	0	0	1	1
13	H	X	H H L H	H	1	0	0	1	0	1	1
14	H	X	H H H L	H	0	0	0	1	1	1	1
15	H	X	H H H H	H	0	0	0	0	0	0	0
BI	X	X	X X X X	L	0	0	0	0	0	0	0
RBI	H	L	L L L L	L	0	0	0	0	0	0	0
LT	L	X	X X X X	H	1	1	1	1	1	1	1

Keterangan: LT = Lamp test

BI = Blanking Input

RBI = Ripple blanking input

RBO = Ripple blanking output

Dari tabel 3-1 dan tampilan 7-segment pada gambar 3-1.b, diatas dapat dibuat program counter 0-9 (decade counter) dengan memberikan batasan/syarat bahwa pada hitungan ke 10 dec / 0a hex program harus kembali ke hitungan awal atau 0, dengan demikian data ke 10 dec (0a hex) sampai dengan data 14 dec (FE hex) tidak pernah ditampilkan pada display 7-segment. Sedangkan data 15 dec (FF hex) ditampilkan ketika program direset atau semua LED pada 7-segment padam. Hal ini juga berlaku untuk program up-counter yang lainnya. Pada program down counter data awalnya adalah data yang terbesar yaitu 9 untuk decade counter dan batasan/syarat untuk kembali ke nilai awal adalah 0 hex - 1 = F hex.

Pada praktikum kali ini kita akan menggunakan semua Port pada DT-51 minimum system baik Port PPI (PA, PB, dan PC) dari IC 82C55 maupun Port 1 IC AT89C51 sebagai jalur input/output.

## PERALATAN :

- Modul DT-51
- PC beserta software
  - Debugger
  - Downloader
  - Editor
- Toggle switch dan 7-Segment sebagai modul I/O

# PROSEDUR :

## Kombinasi Port A, Port B, Port C dan Port1 sebagai Input/Output

### 1. Hardware

#### Input


- Hubungkan konektor (kabel pita) Port A, Port B pada DT-51 ke Modul toggle switch jika input yang digunakan adalah Port A atau Port B. Tetapi bila inputnya adalah Port 1 atau Port C maka *hubungkan konektor (kabel pita) Port 1 atau Port C pada DT-51 ke Modul toggle switch.*

#### Output

- Hubungkan konektor (kabel pita) Port 1, Port C pada DT-51 ke *Modul Display 7-segment* jika output yang digunakan adalah Port 1 atau Port C. Tetapi bila outputnya adalah Port A atau Port B maka hubungkan konektor (kabel pita) Port A atau Port B pada DT-51 ke *Modul Display 7-segment* (lihat gambar 3-3).

#### Selanjutnya

- Hubungkan kabel serial dari konektor DB-9 pada DT-51 ke PC pada Port serial.
- Hubungkan catu daya 9 Volt AC pada modul DT-51.
- Pin Konektor Select pada posisi Pin 1 dan 2 untuk mode *download program*, pada posisi Pin 2 dan 3 untuk mode *stand alone* , setelah download program berhasil.



Gambar 3-3. Koneksi DT-51 MinSys dengan Modul 7-Segment

## 2. Software

### **Editor :**

- Tulislah program pada editor DOS prompt, Notepad atau lainnya.
- Program / file baru tersebut, simpan dalam direktori(folder) DT51 dengan extension *[namafile].ASM*.
- Exit

### **Debugger :**

- Compile file baru dengan perintah `C:\DT51\ASM51 [nama file].ASM` pada DOS Prompt, dengan terlebih dahulu pindah ke direktori DT51.

- Maka akan terbentuk file-file yang berekstension .HEX, .OBJ, dan .LST disamping itu juga muncul pesan, bahwa file yang telah dicompile tersebut terdapat kesalahan atau tidak.
- Bila ada kesalahan, bisa dilihat letak kesalahannya dengan membuka file yang berekstension .LST, melalui C:\DT51\edit.
- Untuk memperbaiki kesalahan (error), buka file yang berekstension .ASM. Jangan lupa disimpan kemudian dicompile kembali.
- Exit

### **Downloader :**

- Untuk mendownload program yang telah sukses dicompile, maka gunakan perintah C:\DT51\DT51L [nama file].HEX
- Apabila proses download selesai, maka akan muncul pesan Download Succeeded.
- Bila gagal, maka akan muncul beberapa prosedur untuk mengecek kembali kesalahan tersebut.

## **PROGRAM PERCOBAAN :**

### **- Program 3.1. Port A Input, Port B Output**

```

$MOD51
;=====
; Program control Counter 00-09 pada display
; 7 Segment menggunakan switch set/reset (H/L)
;=====

        cw      equ    0090h
        portA   equ    2000h
        portB   equ    2001h
        portC   equ    2002h
        addCw   equ    2003h

                org    4000h
                mov    a,#cw
                mov    dptr,#addCw
                movx   @dptr,a
main:      mov    dptr,#portA
                movx   a,@dptr
                cjne  a,#00h,cek
reset:    mov    a,#0ffh
                mov    dptr,#portB
                movx   @dptr,a
                acall delay
                ajmp  main

```

```

cek:    mov    dptr,#portA
        movx  a,@dptr
        cjne a,#01h,main
        mov  a,#00h
lup:    mov    dptr,#portB
        movx @dptr,a
        acall delay
        inc  a
        mov  b,a
        clr  a
        mov  dptr,#portA
        movx a,@dptr
        cjne a,#01h,reset
        clr  a
        mov  a,b
        cjne a,#0ah,lup
        jmp  main

delay:  mov    r1,#03h
del1:   mov    r2,#0ffh
del2:   mov    r3,#0ffh
        djnz  r3,$
        djnz  r2,del2
        djnz  r1,del1
        ret
        end

```

**- Program 3.2. Port 1 Input, Port C Output**

```

$MOD51
;=====
; Program control Counter 00-99 pada display
; 7 Segment menggunakan switch set/reset (H/L)
;=====
        org  4000h
        mov  a,#80h
        mov  dptr,#2003h
        movx @dptr,a

main:   mov  10h,#0ah
        mov  20h,#0h
        mov  b,#10h

        jb  p1.0,run
        jnb p1.0,reset
run:    clr  a
lup:    mov  dptr,#2002h
        movx @dptr,a
        acall delay
        inc  a
        jnb p1.0,reset
        cjne a,10h,lup
        mov  a,10h
        add  a,b

```

```

                cjne a,#0aah,ok
                jmp  main
ok:             mov  10h,a
                mov  a,20h
                add  a,b
                mov  20h,a
                jmp  lup
reset:          mov  a,#0ffh
                mov  dptr,#2002h
                movx @dptr,a
                acall delay
                ajmp main

delay:          mov  r7,#03h
del1:           mov  r6,#0ffh
del2:           mov  r5,#0ffh
                djnz r5,$
                djnz r6,del2
                djnz r7,del1
                ret
                end

```

**- Program 3.3. Port C Input, Port 1 Output**

```

$MOD51
;=====
; Program control Down Counter 09-00 pada display
; 7 Segment menggunakan switch set/reset (H/L)
;=====
                cw      equ  0089h
                portA   equ  2000h
                portB   equ  2001h
                portC   equ  2002h
                addCw   equ  2003h

                org    4000h
                mov    a,#cw
                mov    dptr,#addCw
                movx   @dptr,a
start:          mov    dptr,#portC
                movx   a,@dptr
                cjne  a,#00h,cek
reset:          mov    a,#0ffh
                mov    p1,a
                acall  delay
                ajmp  start
cek:            mov    dptr,#portC
                movx   a,@dptr
                cjne  a,#01h,start

                mov    a,#09h
lup:            mov    p1,a
                acall  delay
                dec    a

```



```

        mov  b,a
        clr  a
        mov  dptr,#portC
        movx a,@dptr
        cjne a,#01h,reset
        clr  a
        mov  a,b
        cjne a,#0ffh,lup
        jmp  start

delay:  mov  r1,#03h
del1:   mov  r2,#0ffh
del2:   mov  r3,#0ffh
        djnz r3,$
        djnz r2,del2
        djnz r1,del1
        ret
        end

```

**- Program 3.4. Port 1 Input, Port C Output**

```

$MOD51
;=====
; Program control Down Counter 99-00 pada display
; 7 Segment menggunakan switch set/reset (H/L)
;=====
        org  4000h
        mov  a,#80h
        mov  dptr,#2003h
        movx @dptr,a

start:  mov  10h,#8fh
        mov  20h,#99h
        mov  b,#10h
        jb   p1.0,run
        jnb  p1.0,reset

run:    mov  a,20h
lup:    mov  dptr,#2002h
        movx @dptr,a
        acall delay
        dec  a
        jnb  p1.0,reset
        cjne a,10h,lup
        mov  a,10h
        clr  c
        subb a,b
        mov  10h,a
        mov  a,20h
        clr  c
        subb a,b
        cjne a,#0f9h,ok
        jmp  start
ok:     mov  20h,a

```

```

                jmp    lup

reset:  mov    a,#0ffh
        mov    dptr,#2002h
        movx  @dptr,a
        acall delay
        jmp    start

delay:  mov    r7,#03h
del1:   mov    r6,#0ffh
del2:   mov    r5,#0ffh
        djnz  r5,$
        djnz  r6,del2
        djnz  r7,del1
        ret
        end

```

## TUGAS :

1. Buat program up-down counter, gunakan Port 1 At89C51 sebagai Input dan Port B PPI sebagai output. Jika Port 1 = 01h, up-counter (00-09) akan dilaksanakan dan bila Port 1 = 00h, down counter (09-00) dilaksanakan.
2. Buat program down counter 35-00, gunakan Port C PPI sebagai input dan Port 1 At89C51 sebagai output. Jika Port C = 01h, counter akan dilaksanakan dan bila Port C = 00h program akan direset.
3. Buat program up-counter 00-50, gunakan Port A PPI sebagai input dan Port 1 At89C51 sebagai output. Jika Port A = 01h, counter akan dilaksanakan dan bila Port A = 00h program akan direset.
4. Buat program down-counter 50-00, gunakan Port A PPI sebagai input dan Port 1 At89C51 sebagai output. Jika Port A = 01h, counter akan dilaksanakan dan bila Port A = 00h program akan direset.